# プログラムのページ

担当 吉 沢 正

## 7007 奇関数の整多項式の逆転について

戸田英雄（通産省電子技術総合研究所プログラム研究室）

### 1. 問 題

$$y=x+a_3x^3+a_5x^5+\cdots+a_mx^m \qquad (1)$$

$m$は奇数とする.

$a_k(k=3(2)m)$ を与えて(1)式を逆転し（逆関数を求める）

$$x=y+b_3y^3+b_5y^5+\cdots+b_my^m \qquad (2)$$

の形で, $b_k(k=3(2)m)$ を決める問題である.

### 2. 算 法

$b_k(k=3(2)m)$ は次式で求められる：

$$b_k=\sum_{n=1}^{\frac{k-1}{2}}(-1)^n(k+n-1)^{(n-1)}\sum_{\beta=k+n-1}^{\alpha=n}\left[\frac{a^l}{l!}\right] \qquad (3)$$

ここで $\sum_{\beta=k+n-1}^{\alpha=n}$ は $\left[\frac{a_3^{m_3}}{m_3!}\cdot\frac{a_5^{m_5}}{m_5!}\cdot\frac{a_7^{m_7}}{m_7!}\cdots\right]$ の積において

$$\alpha=m_3+m_5+m_7+\cdots$$

$$\beta=3m_3+5m_5+7m_7+\cdots$$

で, あらゆる組合せの総和の意味である.

ただし

$$(k+n-1)^{(n-1)}$$
$$=(k+n-1)(k+n-2)\cdots(k+n-1)$$
$$=\frac{(k+n-1)!}{k!}$$

たとえば

$$b_3=-\frac{a_3}{1!}$$

$$b_5=-\frac{a_5}{1!}+6^{(1)}\frac{a_3^2}{2!}$$

$$b_7=-\frac{a_7}{1!}+8^{(1)}\frac{a_3}{1!}\frac{a_5}{1!}-9^{(2)}\frac{a_3^3}{3!}$$

$$b_9=-\frac{a_9}{1!}+10^{(1)}\left(\frac{a_3}{1!}\frac{a_7}{1!}+\frac{a_5^2}{2!}\right)$$
$$-11^{(2)}\frac{a_3^2}{2!}\frac{a_5}{1!}+12^{(3)}\frac{a_3^4}{4!}$$

$$b_{11}=-\frac{a_{11}}{1!}+12^{(1)}\left(\frac{a_3}{1!}\frac{a_9}{1!}+\frac{a_5}{1!}\frac{a_7}{1!}\right)$$
$$-13^{(2)}\left(\frac{a_3^2}{2!}\frac{a_7}{1!}+\frac{a_3}{1!}\frac{a_5^2}{2!}\right)$$
$$+14^{(3)}\frac{a_3^3}{3!}\frac{a_5}{1!}-15^{(4)}\frac{a_3^5}{5!}$$

### 3. プログラムの説明

（1） 入力：$m$ と $a_k(k=3(2)m)$

（2） 出力： $b_k(k=3(2)m)$

（3） プログラム言語：IBM 360/75—PL/I FORMAC

（4） プログラムで用いた変数名：

| 式で用いた記号 | プログラム中の変数名 |
|---|---|
| $m$ | NO |
| $a_3, a_5, \cdots, a_m$ | A(3), A(5), …, A(NO) |
| $b_3, b_5, \cdots, b_m$ | B(3), B(5), …, B(NO) |
| $\sum_\beta^\alpha\left[\frac{a^l}{l!}\right]$ | SUM |
| $(-1)^n(k+n-1)^{(n-1)}\sum_\beta^\alpha\left[\frac{a^l}{l!}\right]$ | BKN |
| $\sum_{n=1}^{(k-1)/2}(-)^n(k+n-1)^{(n-1)}\sum_\beta^\alpha\left[\frac{a^l}{l!}\right]$ | SBKN |

（5） 流れ図：大きな流れは, つぎのようになる.（次ページのプログラムの後にある）

（6） 級数の逆転の部分の説明：

```
DO K=3 TO NO BY 2; ……k=3(2)NO まで
LET(SBKN=0);  ……………b_k を作る準備
  DO N=1 TO (K-1)/2;
  LET (SUM=0); ………… Σ[a^l/l!] を作る準備
  CALL PRTIT (K-1+N, N); (あとで述べる)
  LET (SBKN=SBKN+BKN);
  END;
LET (B("K")=SBKN;
END;
```

PRTIT procedure では $\beta$=K−1+N, $\alpha$=N の値をもらって$\beta$を3以上の奇数 3, 5, 7, … を用いて$\alpha$分割するあらゆる場合を捜し出す. すなわち

プログラムと結果 (1)

```
INPUT TO FORMAC PREPROCESSOR
    ODDINVS: PROCEDURE OPTIONS(MAIN);
             DCL KK(0:100)    BIN FIXED(31,0);
             DCL  P(0:100)    BIN FIXED(31,0);
             DCL( L,K,N,J,MM,NN) BIN FIXED(31,0) ;
        DCL  FX  CHAR(72) ;
 START: ON ENDFILE(SYSIN) GO TO CLOSE ;
         PUT PAGE ;
        KAISU = 0 ;
        GET LIST (NO) ;
        DO L=1 TO NO BY 2 ;
         GET LIST (FX) ;  LET( A("L")="FX" ) ;
        END;
 CHECK: PUT PAGE ;
        DO L=1 TO NO BY 2 ;  PRINT_OUT( A("L")) ; END ;
        DO  K=3 TO  NO BY 2 ;
         LET( SBKN = 0 ) ;
         DO N=1 TO (K-1)/2  ;
          LET( SUM =0 );
          MM= K-1 +N ; NN = N ;
          CALL PARTIT( MM,NN ) ;
          LET( SBKN = SBKN +BKN ); LET( BKN = 0 ) ;
         END ;
         LET ( B("K") = SBKN );
        END ;
        PUT SKIP(2);
        LET( B("1") = 1 ) ;
        DO  K=1 TO NO  BY 2 ;
         PRINT_OUT (B("K")) ;
        END ;
        DO K=1 TO NO BY 2 ;
         LET( A("K") = B("K")   ) ;
        END;
        KAISU = KAISU+1 ;
        IF KAISU =2 THEN GO TO START ;
        GO TO CHECK;
  TERM: PROCEDURE(K,N);
        DCL(K,N,KN1) BIN FIXED(31,0) ;
        DCL(L,PL,KL) BIN FIXED(31,0) ;
        LET( U=1 );
        DO  L=1 TO N ;
         KL = KK(L);
         LET( U= U* A("KL")) ;
        END;
        DO  L=1 TO 50 ;
         PL =P(L) ;
         IF PL<= 1 THEN GO TO SKIP ;
         LET( U= U/FAC("PL")) ;
   SKIP:END ;
        LET(SUM =SUM +U );
        KN1= K+N-1 ;
        LET( BKN = FAC("KN1") /FAC("K") * SUM ) ;
        N2 = N/2 ;
        IF N2  *2 =N  THEN GO TO CONTINUE ;
        LET( BKN = -BKN) ;
CONTINUE:
        END TERM;
    PARTIT: PROCEDURE(MM,NN ) ;
            DCL(M,N,MM,NN )     BIN FIXED(31,0) ;
            DCL(I,J,S,JJ,KJ,II) BIN FIXED(31,0) ;
```

プログラムと結果（2）

```
         M = MM ; N = NN ;
         KK(1) = 1 ;
         DO I=2 TO N ; KK(I)=3 ; END ;
         J=0;
L51:     I=J;  KK(I+1)= KK(I+1)+2 ;
         IF I=0 THEN GO TO L53 ;
L52:     KK(I)=KK(I+1);  I=I-1 ;  IF I=0 THEN GO TO L53 ;
         GO TO L52 ;
L53:     S=0 ;
         DO I=1 TO N ; S=S+ KK(I); END;
         IF M=S THEN GO TO L22;
         IF M<S THEN GO TO L21;
         IF M>S THEN GO TO L23;
    L22:
            DO I=1 TO  50 ; P(I)=0 ;  END;
            DO JJ=1 TO  N ; KJ=KK(JJ);P(KJ)=P(KJ)+1 ; END;
            CALL TERM(K,N);
L21:     J=J+1; GO TO L56;
L23:     J=0;
L56:     IF J ¬= N THEN GO TO L51;
            END PARTIT;
   CLOSE:  CLOSE FILE(SYSPRINT);
  END CODINVS;
```

$$KK(1)+KK(2)+\cdots+KK(\alpha)=\beta$$

を満たす場合をすべて求める．たとえば

$$\beta=K-1+N=9-1+2=10$$

$$\alpha=N=2$$

のときは，10 を 3 以上の奇数で 2 分割するすべての場合で 10=3+7 と 10=5+5 でこれ以外はない．

このとき KK(1)，KK (2) の表と，P(KK(1))，P(KK(2)) の表を作る，いまの例では下記のようになる．

| KK (1) | KK (2) |
|---|---|
| 3 | 7 |
| 5 | 5 |

| P (3) | P (5) | P (7) |
|---|---|---|
| 1 | 0 | 1 |
| 0 | 2 | 0 |

PARTIT procedure の中で TERM procedure を呼んで

$$SUM=A(3)\,A(7)+\frac{A(5)A(5)}{P(5)!}$$

$$=a_3\cdot a_7+\frac{a_5^2}{2!}$$

$$BKN=\frac{(K+N-1)!}{K!}SUM=10\left\{a_3\cdot a_7+\frac{a_5^2}{2!}\right\}$$

の値をそれぞれ SUM と BKN に入れる．




流れ図：大きな流れは上のようになる

## 4. 数値例

① $y=\tan^{-1}x=x-\frac{1}{3}x^3+\frac{1}{5}x^5-\cdots$

を与えて

$$x=\tan y=y+\frac{1}{3}y^3+\frac{2}{15}y^5+\cdots$$

を求める．また，この逆を行ないチェック（計算機からの出力は省略）をした．

$x=\tan y$ の展開式の係数は，

$$B(1)=1,\ B(3)=\frac{1}{3},\ B(5)=\frac{2}{15},\ \cdots$$

$$B(27)=\frac{8\ 37464\ 35170\ 10684}{12\ 98054\ 39119\ 55776\ 40625}$$

と求められるが，これらは文献 2）の表から求めたものと，一致していることを確かめた．

② $y=\sin^{-1}x=x+\frac{1}{2}\frac{1}{3}x^3+\frac{1}{2}\frac{3}{4}\frac{1}{5}x^5+\cdots$

を与えて

$$x=\sin y=y-\frac{1}{3!}y^3+\frac{1}{5!}y^5-\cdots$$

を求める．また，この逆を行ないチェック（出力は略）をした．

数値例として試みた結果を示す．

計算時間は（チェックを含む）

| | | | |
|---|---|---|---|
| FORMAC | STEP | TIME | 21 sec |
| PLIFC | STEP | TIME | 24 sec |
| LKED | STEP | TIME | 25 sec |
| GO | STEP | TIME | 52 sec |
| | TOTAL | TIME | 122 sec |

である．

**数値例①** $y=\tan x^{-1}$ の展開式の係数 $A(i)$ を与えて，$x=\tan y$ の展開式の係数 $B(i)$ を定める場合の出力

```
A(1) = 1                B(1) = 1
A(3) =  - 1/3           B(3) = 1/3
A(5) = 1/5              B(5) = 2/15
A(7) =  - 1/7           B(7) = 17/315
A(9) = 1/9              B(9) = 62/2835
A(11) =  - 1/11         B(11) = 1382/155925
A(13) = 1/13            B(13) = 21844/6081075
A(15) =  - 1/15         B(15) = 929569/638512875
A(17) = 1/17            B(17) = 6404582/10854718875
A(19) =  - 1/19         B(19) = 443861162/1856156927625
A(21) = 1/21            B(21) = 18888466084/194896477400625
A(23) =  - 1/23         B(23) = 113927491862/2900518163668125
A(25) = 1/25            B(25) = 58870668456604/3698160658676859375
A(27) =  - 1/27         B(27) = 8374643517010684/1298054391195577640625
```

**数値例②** $y=\sin^{-1}x$ の展開式の係数 $A(i)$ を与えて，$x=\sin y$ の展開式の係数 $B(i)$ を求める場合の出力

| | |
|---|---|
| A(1) = 1 | B(1) = 1 |
| A(3) = 1/6 | B(3) = - 1/6 |
| A(5) = 3/40 | B(5) = 1/120 |
| A(7) = 5/112 | B(7) = - 1/5040 |
| A(9) = 35/1152 | B(9) = 1/362880 |
| A(11) = 63/2816 | B(11) = - 1/39916800 |
| A(13) = 231/13312 | B(13) = 1/6227020800 |
| A(15) = 143/10240 | B(15) = - 1/1307674368000 |
| A(17) = 6435/557056 | B(17) = 1/355687428096000 |
| A(19) = 12155/1245184 | B(19) = - 1/121645100408832000 |
| A(21) = 46189/5505024 | B(21) = 1/51090942171709440000 |

## 5. 検 討

このプログラムは(3)式の計算に PL/I-FORMAC を使用し，二三の数値例を試みたものである．以下それについての検討を述べる．

1) このプログラムは PL/I-FORMAC の機能の一部分，多倍長の整数計算 (2,295 桁まで可能) だけを利用してある．したがって，一般的な FORMAC のプログラム例にはならない．

2) (3)式を用いないで，数式処理的な FORMAC の機能を十分に利用するやり方では，たとえば，(2)式を(1)式に代入して未定係数法で定めていくようなやり方では，$m$が大きいと時間が非常にかかり実用にならなかった．

3) (3)式を用いて FORTRAN や ALGOL でもプログラムが書けるが，まるめの誤差がはいるので，気持がわるい (文献 1) を見よ)．

4) このプログラムの中心は，$\alpha$を与えて3以上の奇数を用いて$\beta$個に分割するすべての場合を作る procedure である．一松氏やその他の数学者による立派なアルゴリズムもあって筆者の procedure PARTIT が最善とは思わない．

### 参 考 文 献

1) 山内二郎，戸田英雄：級数の逆転の一般式とそのプログラム，第8回プログラミングシンポジウム報告集 (1967)．

2) D. E. Knuth & T. J. Buckholtz: "Computation of Tangent, Euler, and Bernoulli Numbers", Math. of Computation, Vol. 21 (1967), p. 672.

(昭和 45 年 7 月 28 日受付)